

# Advent of Code 2021 - Day 14:

## Extended Polymerization

---

### Parsing

In[61]:=

```
input = "NNCB

CH→B
HH→N
CB→H
NH→C
HB→C
HC→B
HN→C
NN→C
BH→H
NC→B
NB→B
BN→B
BB→N
BC→B
CC→N
CN→C";
```

In[63]:=

```
lines = StringSplit[input, "\n"]
```

Out[63]= { NNCB, , CH→B, HH→N, CB→H, NH→C, HB→C, HC→B, HN→C, NN→C, BH→H, NC→B, NB→B, BN→B, BB→N, BC→B, CC→N, CN→C }

In[199]:=

```
polymertemplate = Characters@lines[[1]]
```

Out[199]= { N, N, C, B }

In[65]:=

```
stringrules = lines[[3 ;;]]
```

Out[65]= { CH→B, HH→N, CB→H, NH→C, HB→C, HC→B, HN→C, NN→C, BH→H, NC→B, NB→B, BN→B, BB→N, BC→B, CC→N, CN→C }

```
In[81]:= rules = Map[(Characters@StringTake[#, 2] → {{StringTake[#, {1}], StringTake[#, {4}]},
  {StringTake[#, {4}], StringTake[#, {2}]}}) &, stringrules] // Sort
```

```
Out[81]= {{B, B} → {{B, N}, {N, B}}, {B, C} → {{B, B}, {B, C}},
  {B, H} → {{B, H}, {H, H}}, {B, N} → {{B, B}, {B, N}},
  {C, B} → {{C, H}, {H, B}}, {C, C} → {{C, N}, {N, C}}, {C, H} → {{C, B}, {B, H}},
  {C, N} → {{C, C}, {C, N}}, {H, B} → {{H, C}, {C, B}}, {H, C} → {{H, B}, {B, C}},
  {H, H} → {{H, N}, {N, H}}, {H, N} → {{H, C}, {C, N}}, {N, B} → {{N, B}, {B, B}},
  {N, C} → {{N, B}, {B, C}}, {N, H} → {{N, C}, {C, H}}, {N, N} → {{N, C}, {C, N}}}
```

## Create Matrix of Pair Insertion

```
In[87]:= intmapping = MapIndexed[#1 → First@#2 &, rules[[All, 1]]]
```

```
Out[87]= {{B, B} → 1, {B, C} → 2, {B, H} → 3, {B, N} → 4, {C, B} → 5,
  {C, C} → 6, {C, H} → 7, {C, N} → 8, {H, B} → 9, {H, C} → 10, {H, H} → 11,
  {H, N} → 12, {N, B} → 13, {N, C} → 14, {N, H} → 15, {N, N} → 16}
```

```
In[92]:= {"B", "B"} → {{{"B", "N"}, {"N", "B"}}} /. intmapping
```

```
Out[92]= 1 → {4, 13}
```

```
In[110]:= transformations =
  Map[{{#[[2, 1]], #[[1]]}, {#[[2, 2]], #[[1]]}} &, rules] /. intmapping //
  Flatten[#, 1] &
```

```
Out[110]= {{4, 1}, {13, 1}, {1, 2}, {2, 2}, {3, 3}, {11, 3}, {1, 4}, {4, 4},
  {7, 5}, {9, 5}, {8, 6}, {14, 6}, {5, 7}, {3, 7}, {6, 8}, {8, 8}, {10, 9},
  {5, 9}, {9, 10}, {2, 10}, {12, 11}, {15, 11}, {10, 12}, {8, 12}, {13, 13},
  {1, 13}, {13, 14}, {2, 14}, {14, 15}, {7, 15}, {14, 16}, {8, 16}}
```

```
In[120]:= matrix = SparseArray[
  transformations → Table[1, Length@transformations], {Length@rules, Length@rules}]
```

```
Out[120]= SparseArray[ Specified elements: 32  
Dimensions: {16, 16}]
```

```
In[153]:= MatrixForm[matrix]
Out[153]//MatrixForm=
```

$$\begin{pmatrix} 0 & 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

## Initial Polymer

```
In[138]:= Rule@@# & /@ (Tally@Partition[Characters[lines[[1]]], 2, 1] /. intmapping)
Out[138]= {16 → 1, 14 → 1, 5 → 1}
```

```
init = Table[0, Length@rules] //
  ReplacePart[Rule@@# & /@ (Tally@Partition[polymertemplate, 2, 1] /. intmapping)]
```

```
Out[140]= {0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 1}
```

## Multiplication

```
In[200]:= resultingpairs = MatrixPower[matrix, 10, init] * rules[[All, 1]]
doubletotal =
  Total@Flatten@resultingpairs + First@polymertemplate + Last@polymertemplate
```

```
Out[200]= {{812 B, 812 B}, {120 B, 120 C}, {81 B, 81 H}, {735 B, 735 N}, {115 C, 115 B},
  {60 C, 60 C}, {21 C, 21 H}, {102 C, 102 N}, {26 H, 26 B}, {76 H, 76 C},
  {32 H, 32 H}, {27 H, 27 N}, {796 N, 796 B}, {42 N, 42 C}, {27 N, 27 H}, {0, 0}}
```

```
Out[201]= 3498 B + 596 C + 322 H + 1730 N
```

```
In[202]:= #[[1]] / 2 & /@ List@@doubletotal // MinMax // Differences // First
```

```
Out[202]= 1588
```